Le triangle et le tapis de Sierpinski

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Dans le cadre des séries géométriques, cette feuille Maple a été développée pour les illustrations d'un exercice de mon cahier de notes pour le cours EED. La présente feuille est une version documentée. La lecture de cette feuille de travail est l'occasion pour le lecteur d'aborder la notion de boucle informatique et pour constater la possibilité d'inclure une procédure locale dans une procédure.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

La géométrie fractale permet de caractériser des objets ayant une forme irrégulière posédant la propriété d'invariance par changement d'échelle. De manière simplifiée, on définit un objet fractal comme un objet ayant une homothétie interne, c'est-à-dire qu'une portion de l'objet est identique à l'objet complet. Si vous regardez un objet fractal en le grossissant régulièrement, vous allez voir encore la même forme. On dit d'un objet fractal qu'il possède la particularité d'autosimilarité.

Le terme "fractal" vient du latin, "fractus" qui désigne un objet fracturé, de forme très irrégulière. C'est Benoît Mandelbrot qui, en 1975, a introduit ce terme pour désigner ces fameux objets mathématiques.

Le triangle de Sierpinski (1882-1969) est un exemple de fractal de Sierpinski. Le fractal de Sierpinski a été étudié par Sierpinski en 1915. Cet objet peut être construit en réalisant l'algorithme suivant:

- Étape 1: tracer un triangle équilatéral plein de côté a;
- Étape 2: subdiviser ce triangle en quatre triangles équilatéraux congruents puis enlever le triangle central;
- Étape k, $k = 3, 4, \ldots$: répéter indéfiniment cette construction sur chaque triangle plein restant.



Initialisation

```
> restart;
  with(plots,display,polygonplot,setoptions):
  setoptions(size=[300,300]);
```

Le triangle de Sierpinski

Voici une procédure qui dessine la *n*-ième etape de la construction du triangle de Sierpinski, a partir du triangle de sommets A, B, C (pas nécessairement équilatéral):

```
> Sierpinski:=proc(étapes,A,B,C)
  local i,j,h,L,M,Opts,T;
  Opts:=[args[5..nargs]];
  h:=proc(P,lambda,V)
   lambda*(V-P)+P;
  end proc:
  L:=[[A,B,C]];   #liste contenant le triangle initial
  for i from 1 to étapes do
   M:=[];
     for j in L do
    M:=[map(a->h(j[1],1/2,a),j),map(a->h(j[2],1/2,a),j),map(a->h(j[3],
  1/2,a),j),op(M)];
     od;
   L:=M;
  od;
  T:=polygonplot(L,axes=none,op(Opts));
  display(T,op(Opts));
  end:
```

```
> seq(print(Sierpinski(k,[0,0],[2,0],[1,2],color=plum,size=[200,200])),
  k=0..5);
```













```
> Sierpinski(4,[0,0],[2,0],[1,2],color=magenta);
```



Alons-y d'une animation illustrant les 6 premières étapes.

```
>  display(seq(Sierpinski(k,[0,0],[2,0],[1,1],color=orange),k=0..5),
  insequence=true);
```



Le tapis de Sierpinski

Le tapis de Sierpinsky peut être construit en réalisant l'algorithme suivant:

- Étape 1: tracer un carré plein de côté a;
- Étape 2: subdiviser ce carré en neuf carrés congruents puis enlever le carré central;
- Étape k, $k = 3, 4, \ldots$: répéter indéfiniment cette construction sur chaque carré plein restant.

```
> Tapis:=proc(étapes,A,B,C,D)
  local i,j,homot,L,M,Opts,T;
  Opts:=[args[6..nargs]];
  homot:=proc(P,lambda,V)
   lambda*(V-P)+P;
  end proc:
  L:=[[A,B,C,D]];
  for i from 1 to étapes do
   M:=[];
    for j in L do
    M:=[map(a->homot(j[1],1/3,a),j),
    map(a->homot((j[1]+j[2])/2,1/3,a),j),
    map(a->homot(j[2],1/3,a),j),
    map(a->homot((j[2]+j[3])/2,1/3,a),j),
    map(a->homot(j[3],1/3,a),j),
    map(a->homot((j[3]+j[4])/2,1/3,a),j),
    map(a->homot(j[4],1/3,a),j),
    map(a->homot((j[4]+j[1])/2,1/3,a),j),
    op(M)];
   od:
  L:=M;
  od;
  T:=polygonplot(L,axes=none,scaling=constrained,op(Opts));
  display(T,op(Opts));
  end:
```

```
> seq(print(Tapis(k,[0,0],[2,0],[2,2],[0,2],color = ["Bright
  GreenishYellow"],size=[200,200])),k=0..3);
```





```
> Tapis(3,[0,0],[2,0],[2,2],[0,2],color = ["Bright Purple"],size=[200,
  200]);
```



Alons-y d'une animation illustrant les 5 premières étapes

```
> display(seq(Tapis(k,[0,0],[2,0],[2,2],[0,2],color=orange),k=0..4),
  insequence=true,size=[200,200])
```



Exercices

No.1 Le triangle de Sierpinski

a) Obtenir le terme général de la PG $\left\{t_k\right\}$ donnant le nombre de triangles enlevés à la k^e itérations.

b) Donner le nombre de triangles enlevés lors de la 10^e itérations.

c) Obtenir le terme général de la PG $\left\{a_k\right\}$ donnant l'aire de la surface enlevée lors de la k^e itérations.

Prendre l'aire du triangle de départ égale à une unité de surface.

d) Déterminer l'aire de la surface enlevée lors de la 10^e itérations.

e) Obtenir le terme général de la PG $\left\{T_k\right\}$donnant le nombre total de triangles enlevés après la k^e itérations.

f) Donner le nombre total de triangles enlevés après la 10^e itérations.

g) Obtenir le terme général de la PG $\left\{A_k\right\}$ donnant l'aire total de la surface enlevée après la k^e itérations.

Prendre l'aire du triangle de départ égale à une unité de surface.

h) Déterminer l'aire total de la surface enlevée après la 10^e itérations.

i) Déterminer l'aire total de la surface enlevée lorsque $k \rightarrow \infty$.

j) Quelle est donc la caractéristique de cet objet fractal en regard de l'aire des triangles en noir lorsque $k \rightarrow \infty$?

No.2 Même questions qu' au No. 1 mais avec le tapis de Sierpinski.